MATHEMATICS NOTES

NOTE 14

AUGUST 1970

TEF

A subroutine for the calculation of the Incomplete Elliptic Integrals of the First and Second Kind.

Terry L. Brown

The Dikewood Corporation

Abstract

This note describes a computer subroutine which calculates the Incomplete Elliptic Integrals $F(\phi|m)$ and $E(\phi|m)$. The routine accepts any value both positive and negative of the amplitude ϕ. The parameter m is restricted to $0 \leq m \leq 1$.

INTRODUCTION

Subroutine TEF is a computer code written in standard ASA FORTRAN IV for the Control Data Corporation 6600 computer at the Air Force Weapons Laboratory at Kirtland Air Force Base, New Mexico. Two completely different methods are employed to calculate the Incomplete Elliptic Integral of the First Kind $F(\phi \mid m)$ and the Second Kind $E(\phi \mid m)$ depending upon the size of the parameter m. Also listed is a routine to calculate the Complete Elliptic Integrals K(m) and E(m).

The incomplete elliptic integrals of the First and Second Kind are defined as follows:

$$F(\phi|m) = \int_0^{\phi} [1 - m \sin^2(\theta)]^{-\frac{1}{2}} \, d\theta$$

$$E(\phi|m) = \int_0^{\phi} [1 - m \sin^2(\theta)]^{\frac{1}{2}} \, d\theta \qquad (1)$$

where ϕ is the amplitude and m is the parameter. The complementary parameter m_1 is defined as

$$m + m_1 = 1 \qquad (2)$$

and

$$m = \sin^2(\alpha) \qquad (3)$$

where α is the modular angle. It should be noted that the incomplete elliptic integrals are written in several different forms. Dependence on the parameter m is denoted by a vertical stroke preceding the parameter as written above. Dependence on the modular angle α is denoted by a backward stroke preceding the modular angle as

$$F(\phi\backslash\alpha), \quad E(\phi\backslash\alpha) \qquad (4)$$

Dependence on the modulus k is denoted in one of two ways as

$$F(\phi, k), \quad E(\phi, k)$$

or

$$F(\phi|k), \quad E(\phi|k) \qquad (5)$$

where

$$m = k^2$$

and

$$m_1 = (k')^2 \qquad (6)$$

Several different forms are used in this note and these forms are generally dictated by the references.

For the purpose of the routine given in this note the parameter m is chosen for the input to the subroutine. The restrictions placed upon m are

$$0 \leq m \leq 1 \qquad (7)$$

The routine accepts amplitudes ϕ of any magnitude using the relationship[1]

$$F(s\pi \pm \phi | m) = 2sK \pm F(\phi | m) \qquad (8)$$

and

$$E(s\pi \pm \phi | m) = 2sE \pm E(\phi | m) \qquad (9)$$

where the complete elliptic integrals K and E are defined as the incomplete elliptic integrals with the amplitude equal to $\frac{\pi}{2}$ as

$$K(m) = K = F(\tfrac{\pi}{2} | m) = \int_0^{\frac{\pi}{2}} [1 - m \sin^2(\theta)]^{-\frac{1}{2}} d\theta \qquad (10)$$

and

$$E(m) = E = E(\tfrac{\pi}{2} | m) = \int_0^{\frac{\pi}{2}} [1 - m \sin^2(\theta)]^{\frac{1}{2}} d\theta \qquad (11)$$

---

[1] Handbook of Mathematical Functions, AMS 55, M. Abramowitz and I. A. Stegun, Editors, National Bureau of Standards, 1964, eqns. 17.4.3 and 17.4.4.

The following relationships are used for negative amplitudes[2]

$$F(-\phi | m) = -F(\phi | m)$$
$$E(-\phi | m) = -E(\phi | m) \qquad (12)$$

The incomplete elliptic integral of the First Kind is difficult to calculate, especially for values of ϕ near $\frac{\pi}{2}$ and m near unity. For this reason TEF calculates $F(\phi | m)$ using two methods depending upon the size of the parameter m. Consequently $E(\phi | m)$ is also computed using two methods.

For values of m less than .75 the following infinite series[3] are used to calculate the incomplete elliptic integral of the First and Second Kinds respectively

$$F(\phi | m) = \int_0^{\phi} [1 - m \sin^2(\theta)]^{-\frac{1}{2}} \, d\theta$$
$$= \frac{2\phi}{\pi} K - \sin(\phi)\cos(\phi) \left[\frac{1}{2} A_2 m + \frac{1 \cdot 3}{2 \cdot 4} A_4 m^2 + \frac{1 \cdot 3 \cdot 5}{2 \cdot 4 \cdot 6} A_6 m^3 + \ldots \right] \qquad (13)$$

$$E(\phi | m) = \int_0^{\phi} [1 - m \sin^2(\theta)]^{\frac{1}{2}} \, d\theta$$
$$= \frac{2\phi}{\pi} E + \sin(\phi)\cos(\phi) \left[\frac{1}{2} A_2 m + \frac{1}{2 \cdot 4} A_4 m^2 + \frac{1 \cdot 3}{2 \cdot 4 \cdot 6} A_6 m^3 + \ldots \right] \qquad (14)$$

---

[2] Reference 1, eqns. 17.4.1 and 17.4.2.

[3] Herbert Bristol Dwight, Tables of Integrals and Other Mathematical Data, The MacMillan Company, 1965, p. 172, eqns. 775 and 777.

where $A_2 = \frac{1}{2}$

$$A_4 = \frac{3}{2\cdot 4} + \frac{1}{4}\sin^2(\phi)$$

$$A_6 = \frac{3\cdot 5}{2\cdot 4\cdot 6} + \frac{5}{4\cdot 6}\sin^2(\phi) + \frac{1}{6}\sin^4(\phi) \tag{15}$$

$$A_8 = \frac{3\cdot 5\cdot 7}{2\cdot 4\cdot 6\cdot 8} + \frac{5\cdot 7}{4\cdot 6\cdot 8}\sin^2(\phi) + \frac{7}{6\cdot 8}\sin^4(\phi) + \frac{1}{8}\sin^6(\phi)$$

with K and E the complete elliptic integrals of the First and Second Kind.

This method although extremely fast requires many iterations, that is, terms (one term per iteration) of the series to produce the required degree of accuracy for m close to 1. For example, for $\phi = 85^\circ$ and $m = \sin^2(88^\circ)$ approximately 20,000 iterations were required to make the last term less than 10^{-14}. Consequently a second method is employed to produce $F(\phi \mid m)$ for values of m close to unity. To calculate $F(\phi \mid m)$ and $E(\phi \mid m)$ the descending Landen transformation[4] is used where ϕ_1, ϕ_2, ..., ϕ_N are successively determined from

$$\tan(\phi_{n+1} - \phi_n) = \frac{b_n}{a_n}\tan(\phi_n), \quad n=0, 1, 2, \ldots N \tag{16}$$

where $\phi_o = \phi$.

Then to the accuracy desired[5]

$$F(\phi \backslash \alpha) = \frac{\phi_N}{2^N a_N} \tag{17}$$

and

$$E(\phi \backslash \alpha) = \frac{E}{K} F(\phi \backslash \alpha) + c_1 \sin(\phi_1) + c_2 \sin(\phi_2) + \ldots + c_N \sin(\phi_N) \tag{18}$$

---

[4]Reference 1, eqn. 17.6.8.

[5]Reference 1, eqns. 17.6.9 and 17.6.10.

where a_n and b_n are determined by the process of the Arithmetic-Geometric Mean[6]. Starting with

$$a_o = 1, \; b_o = \cos(\alpha), \quad c_o = \sin(\alpha) \tag{19}$$

a_n, b_n, and c_n can then be determined as

$$a_n = \frac{a_{n-1} + b_{n-1}}{2}$$

$$b_n = (a_{n-1} \cdot b_{n-1})^{\frac{1}{2}}$$

$$c_n = \frac{1}{2}(a_{n-1} - b_{n-1})$$

where $0 \leq |c_n| \leq \epsilon$ to the degree of accuracy ϵ specified.

It might be noted that $E(\phi|m)$ can be accurately calculated using any one of several methods including Gaussian Quadrature which would essentially reproduce the tables of the incomplete elliptic integrals in reference 1. However, since most of the logic to compute $E(\phi|m)$ is contained in the logic for $F(\phi|m)$ the remaining logic was also added.

It can be shown that the summation given in equations 13 and 14 above and by the ratio test[7] that

$$|R_n| \leq \frac{|a_n^2|}{|a_n| - |a_{n+1}|} \tag{20}$$

This test is used to determine the number of terms of the summation that are to be added to achieve the desired degree of accuracy.

---

[6] Reference 1, eqns. 17.6.1 and 17.6.2.

[7] Wilfred Kaplan, Advanced Calculus, Addison-Wesley Publishing Company, Reading, Massachusetts, 1952.

The discrepancies between the values of $F(\phi\,|\,m)$ and $E(\phi\,|\,m)$ given in reference 1 and the values returned from subroutine TEF are given in Tables 1 and 2. All of the values differ by no more than 10^{-8}. The values of $F(\phi\,|\,m)$ and $E(\phi\,|\,m)$ were specially calculated by both methods in subroutine TEF as a test of the numerical methods involved. The series method and the method using the Landen transformation returned the same values of $F(\phi\,|\,m)$ and $E(\phi\,|\,m)$ when run in double precision. This fact leads one to doubt the accuracy of the tables[8] of the incomplete elliptic integrals as given in the Handbook of Mathematical Functions. To avoid any error which might have been introduced by the routine that returns the complete elliptic integrals, any value of K (m) or E(m) that did not agree to 10^{-15} with the table in the AMS 55 was explicitly entered.

Subroutine TEF tests for special values of the parameter m and the amplitude ϕ to avoid needless calculations. These special values and the method[9] by which they are computed are as follows

$$F(\phi\,|\,1) = \ln\left[\tan\left(\frac{\pi}{4} + \frac{\phi}{2}\right)\right], \quad E(\phi\,|\,1) = \sin(\phi) \tag{21}$$

$$F(\phi\,|\,0) = \phi, \quad E(\phi\,|\,0) = \phi \tag{22}$$

$$F\left(\frac{\pi}{2}\,\middle|\,m\right) = K(m), \quad E\left(\frac{\pi}{2}\,\middle|\,m\right) = E(m) \tag{23}$$

$$F\left(\frac{\pi}{2}\,\middle|\,1\right) = 10^{75} \ (\infty), \quad E\left(\frac{\pi}{2}\,\middle|\,1\right) = 1 \tag{24}$$

$$F(0\,|\,0) = 0, \quad E(0\,|\,0) = 0 \tag{25}$$

The subroutine TEK is called by TEF to compute the complete elliptic integrals E(m) and K(m) accurate to about 10^{-12}. Two methods

---

[8] Reference 1, pp. 613 - 618, Tables 17.5 and 17.6.

[9] Reference 1, p. 594, eqns. 17.4.19, 17.4.21, 17.4.23 and 17.4.25.

Table 1.

Comparison of values of $F(\phi \backslash \alpha)$ as given in the Handbook of Mathematical Functions and values returned from subroutine TEF

$F(\phi \backslash \alpha)$	value listed	computed value
$F(5° \backslash 48°)$	0.08732765	0.08732766
$F(10° \backslash 58°)$	0.17517260	0.17517259
$F(10° \backslash 62°)$	0.17522690	0.17522691
$F(10° \backslash 86°)$	0.17542143	0.17542142
$F(15° \backslash 44°)$	0.26324404	0.26324403
$F(15° \backslash 46°)$	0.26335019	0.26335020
$F(20° \backslash 70°)$	0.35547959	0.35547958
$F(20° \backslash 82°)$	0.35622881	0.35622880
$F(25° \backslash 28°)$	0.43932365	0.43932364
$F(25° \backslash 48°)$	0.44404397	0.44404396
$F(25° \backslash 74°)$	0.44967538	0.44967539
$F(30° \backslash 80°)$	0.54842535	0.54842534
$F(35° \backslash 50°)$	0.63363947	0.63363946
$F(35° \backslash 52°)$	0.63511150	0.63511149
$F(35° \backslash 64°)$	0.64351521	0.64351520
$F(35° \backslash 78°)$	0.65067415	0.65067414
$F(35° \backslash 84°)$	0.65228622	0.65228621
$F(50° \backslash 72°)$	0.99163507	0.99163506
$F(55° \backslash 86°)$	1.15261652	1.15261651
$F(60° \backslash 50°)$	1.16431637	1.16431636
$F(60° \backslash 56°)$	1.19275650	1.19275649
$F(60° \backslash 60°)$	1.21259661	1.21259662
$F(60° \backslash 84°)$	1.31117166	1.31117165
$F(70° \backslash 56°)$	1.45726935	1.45726934
$F(75° \backslash 46°)$	1.49668437	1.49668438
$F(75° \backslash 82°)$	1.97316666	1.97316665
$F(80° \backslash 82°)$	2.31643897	2.31642896
$F(85° \backslash 56°)$	1.90143591	1.90143590
$F(85° \backslash 66°)$	2.13070052	2.13070051

Table 2.

Comparison of values of $E(\phi\backslash\alpha)$ as given in the Handbook of Mathematical Functions and values returned from subroutine TEF

$E(\phi\backslash\alpha)$	value listed	computed value
$E(10^\circ\backslash 70^\circ)$	0.17375210	0.17375209
$E(15^\circ\backslash 68^\circ)$	0.25924104	0.25924103
$E(15^\circ\backslash 48^\circ)$	0.26016110	0.26016109
$E(20^\circ\backslash 74^\circ)$	0.34256478	0.34256479
$E(25^\circ\backslash 74^\circ)$	0.42368913	0.42368914
$E(30^\circ\backslash 84^\circ)$	0.50026923	0.50026922
$E(30^\circ\backslash 74^\circ)$	0.50186633	0.50186634
$E(35^\circ\backslash 72^\circ)$	0.57733641	0.57733640
$E(35^\circ\backslash 38^\circ)$	0.59723431	0.59723432
$E(40^\circ\backslash 20^\circ)$	0.69206954	0.69206953
$E(45^\circ\backslash 48^\circ)$	0.74409773	0.74409772
$E(50^\circ\backslash 54^\circ)$	0.80601230	0.80601229
$E(55^\circ\backslash 46^\circ)$	0.89246858	0.89246857
$E(60^\circ\backslash 64^\circ)$	0.90689460	0.90689461
$E(70^\circ\backslash 58^\circ)$	1.03614663	1.03614664
$E(75^\circ\backslash 82^\circ)$	0.97598331	0.97598330
$E(75^\circ\backslash 76^\circ)$	0.99517606	0.99517605
$E(75^\circ\backslash 70^\circ)$	1.02171634	1.02171633
$E(80^\circ\backslash 30^\circ)$	1.31605841	1.31605840
$E(85^\circ\backslash 72^\circ)$	1.07377505	1.07377504
$E(85^\circ\backslash 6^\circ)$	1.47970717	1.47970716

are used in the subroutine depending on the size of m. If $0 \leq m \leq (1 - 10^{-5})$ the following sequence [10, 11] is generated

$$k_0 = k$$

$$k_{n+1} = \frac{1 - k'_n}{1 + k'_n} \qquad (26)$$

where the complement modulus k' is given by

$$k'_n = (1 - k_n^2)^{\frac{1}{2}} \qquad n = 0,\ 1,\ 2, \ldots r \qquad (27)$$

with r being determined from the relation

$$k_r < 10^{-15}$$

Since $k_n \to 0$ as $n \to \infty$ K(m) and E(m) can be obtained from

$$K_n = \frac{\pi}{2} \quad , \quad E_n = \frac{\pi}{2}$$

$$K_{n-1} = \frac{2K}{1 + k'_n} \quad , \quad E_{n-1} = (1 + k_{n-1}) E_n - \frac{2k'_{n-1}}{1 + k'_{n-1}} K_n$$

$$\vdots \qquad\qquad \vdots$$

$$K = K_0 \quad , \quad E = E_0 \qquad (28)$$

Subroutine TEK uses the series[12]

---

[10] J. H. Flinchum and D. E. Amos, AFWL Library Package MATH 13.

[11] G. A. Korn and T. M. Korn, Mathematical Handbook for Scientists and Engineers, McGraw-Hill, 1961, pp 711, eqns. 21.6-6(c).

[12] Reference 3, eqns. 773.3 and 774.3.

$$K = \ln[4(m_1)^{-\frac{1}{2}}] + \frac{1^2}{2^2}\left\{\ln[4(m_1)^{-\frac{1}{2}}] - \frac{2}{1\cdot 2}\right\} m_1$$

$$+ \frac{1^2\cdot 3^2}{2^2\cdot 4^2}\left\{\ln[4(m_1)^{-\frac{1}{2}}] - \frac{2}{1\cdot 2} - \frac{2}{3\cdot 4}\right\} m_1^2$$

$$+ \frac{1^2\cdot 3^2\cdot 5^2}{2^2\cdot 4^2\cdot 6^2}\left\{\ln[4(m_1)^{-\frac{1}{2}}] - \frac{2}{1\cdot 2} - \frac{2}{3\cdot 4} - \frac{2}{5\cdot 6}\right\} m_1^3 + \ldots \quad (29)$$

and

$$E = 1 + \frac{1}{2}\left\{\ln[4(m_1)^{-\frac{1}{2}}] - \frac{1}{1\cdot 2}\right\} m_1$$

$$+ \frac{1^2\cdot 3}{2^2\cdot 4}\left\{\ln[4(m_1)^{-\frac{1}{2}}] - \frac{2}{1\cdot 2} - \frac{1}{3\cdot 4}\right\} m_1^2$$

$$+ \frac{1^2\cdot 3^2\cdot 5}{2^2\cdot 4^2\cdot 6}\left\{\ln[4(m_1)^{-\frac{1}{2}}] - \frac{2}{1\cdot 2} - \frac{2}{3\cdot 4} - \frac{1}{5\cdot 6}\right\} m_1^3$$

$$+ \ldots \quad (30)$$

for $(1 - 10^{-5}) < m < 1$. However, when m is close to one the number of significant bits lost in the subtraction (to obtain m_1 as $m_1 = 1-m$) can reach the word length of the computer. Subsequently, subroutine TEK has been written with the option of supplying m_1 directly to the routine thereby avoiding this loss of significant bits. This procedure works provided of course m_1 can be calculated in the calling routine other than by subtraction of m from 1. To exercise this option the variable ID is set non-zero and m_1 is supplied to TEK instead of m through the variable RM. The parameters EK and E return the value of K(m) and E(m) as always. A listing of subroutine TEK is given in Appendix B.

To use subroutine TEF the calling routine must furnish the standard FORTRAN statement

CALL TEF (PHI, RM, CRIT, F, E)

The first three parameters PHI, RM and CRIT are supplied to TEF and the last two are returned to the calling routine. The parameters and their uses are listed below.

1. PHI — TYPE REAL. This variable corresponds to the amplitude ϕ.

2. RM — TYPE REAL. This variable represents the parameter m.

3. CRIT — TYPE REAL. This is the error criteria for the calculations of $F(\phi | m)$ and $E(\phi | m)$. If m is less than .75 then CRIT determines the number of terms in the series such that the remainder R_n of the series is less than the criteria. If m is greater than .75 the criteria is used to determine the accuracy to which the arithmetic-geometric mean is carried.

4. F — TYPE REAL. This variable contains the value of $F(\phi | m)$ that is returned to the calling routine.

5. E — TYPE REAL. This variable contains the value of $E(\phi | m)$ that is returned to the calling routine.

Subroutine TEK is called from subroutine TEF as follows

CALL TEK (ID, RM, EK, E)

The first two parameters ID and RM are supplied to subroutine TEK and the last two are returned to the calling routine. The use of these parameters is described below.

1. ID TYPE INTEGER. This determines which variable the subroutine expects to receive. If ID = 0 RM $\longleftrightarrow$ m. If ID $\neq$ 0 RM $\longleftrightarrow m_1$.

2. RM REAL. This variable represents the parameter m or its complement m_1.

3. EK TYPE REAL. This variable contains the value of K(m) that is returned to the calling routine.

4. E TYPE REAL. This variable contains the value E(m) that is returned to the calling routine.

An output file is required by subroutine TEF and subroutine TEK for the printed error message from the check on the size of m.

Core storage requirements for the two subroutines TEF and TEK are approximately 1250_8 and 420_8 respectively.

The time to calculate $F(\phi | m)$ and $E(\phi | m)$ varies according to the value of ϕ and m. However, as a guide to the time involved, the tables[8] of $F(\phi \backslash \alpha)$ and $E(\phi \backslash \alpha)$ in the Handbook of Mathematical Functions AMS 55 can essentially be reproduced in approximately 6 seconds.

SUMMARY

Subroutine TEF is a general purpose routine for computing $F(\phi|m)$ and $E(\phi|m)$ with a high degree of accuracy for a wide range of ϕ with the parameter m in the range $0 \leq m \leq 1$. Although not a long routine there was an attempt to increase speed and accuracy at the expense of space in the general trade off between core storage and central processor time. Subroutine TEK is used in the computation of $F(\phi|m)$ and $E(\phi|m)$ but can be used by itself as a general purpose routine.

ACKNOWLEDGEMENTS

The suggestions of Dr. Carl E. Baum of AFWL concerning this routine are gratefully acknowledged as are the many helpful discussions with A1C Robert N. Marks.

Appendix A:

Listing of subroutine TEF

```
      SUBROUTINE TEF (PH1,RM,SFG,TF,TE)
      DATA PIO4/.785398163397448/,TPI/6.28318530717959/
      DATA PI,PIO2/3.141592653589793238462643E0,1.5707963267948966192E0/
      DIMENSION AA(50), BB(50), CC(50), PSAV(50)
      IF (ABS(RM-.5)-.5) 15,15,5
5     PRINT 10, RM
10    FORMAT (5X,9H*********,3X,13HLOOK OUT M = ,F8.3,3X,9H*********)
      RETURN
15    IF (PH1) 20,25,25
20    W=-1.
      PH=-PH1
      GO TO 30
25    W=1.
      PH=PH1
30    RK=SQRT(RM)
      N=PH/TPI
      A=PH-FLOAT(N)*TPI
      B=A/PIO2
      K=B
      NQ=K+1
      GO TO (35,40,45,50), NQ
35    NK=4*N
      SIGNEM=1.
      AP=A
      GO TO 55
40    NK=4*N+2
      SIGNEM=-1.
      AP=PI-A
      GO TO 55
45    NK=4*N+2
      SIGNEM=1.
      AP=A-PI
      GO TO 55
50    NK=4*N+4
      SIGNEM=-1.
      AP=TPI-A
55    CNK=NK
      PHI=AP
      CALL TEK (0,RM,EK,EE)
      PLUS=CNK*EK
      PLUS1=CNK*EE
      IT=0
      IF (ABS(PHI-PIO2)-1.E-10) 60,60,65
60    IT=1
65    IF (ABS(RK-1.E0)-1.E-10) 70,85,85
70    IT=IT+1
      GO TO (75,80), IT
75    TF=W*(PLUS+SIGNEM*ALOG(TAN(PIO4+PHI*.5)))
      TE=W*(PLUS1+SIGNEM*SIN(PHI))
      RETURN
80    TF=W*1.E75
      TE=W*(PLUS1+SIGNEM)
      RETURN
85    IF (ABS(RK)-1.E-15) 90,95,95
90    TF=W*(PLUS+SIGNEM*PHI)
```

```
      TE=W*(PLUS1+SIGNEM*PHI)
      RETURN
95    IT=IT+1
      GO TO (105,100), IT
100   CALL TEK (0,RM,EK,EE)
      TF=W*(PLUS+SIGNEM*EK)
      TE=W*(PLUS1+SIGNEM*EE)
      RETURN
105   IF (ABS(PHI)-1.E-50) 110,115,115
110   TF=W*PLUS
      TE=W*PLUS1
      RETURN
115   IF (RM-.75) 120,140,140
120   CALL TEK (0,RM,EK,EE)
      S=SIN(PHI)
      C=COS(PHI)
      SK=RM
      CE=2.*PHI/PI
      TZ=CE*EK
      T1=CE*EE
      A=.5E0
      T=.5E0*A*SK
      R=T
      SS=S*S
      PS=1.E0
      H=.5
      F=.5E0
      PK=SK
      U1=10.
      DO 130 I=2,20000
      J=I*2
      D=FLOAT(J-1)
      G=FLOAT(J-3)
      E=1./FLOAT(J)
      PS=SS*PS
      A=E*(D*A+PS)
      F=D*E*F
      H=G*E*H
      PK=PK*SK
      U=F*A*PK
      IF (U1*U1/(U1-U)-SIG) 135,135,125
125   U1=U
      T=U+T
130   R=H*A*PK+R
135   TF=W*((TZ-S*C*T)*SIGNEM+PLUS)
      TE=W*((T1+S*C*R)*SIGNEM+PLUS1)
      RETURN
140   ALPHAR=ASIN(RK)
      AA(1)=1.
      BB(1)=COS(ALPHAR)
      DO 145 I=2,50
      II=I-1
      AA(I)=.5*(AA(II)+BB(II))
      BB(I)=SQRT(AA(II)*BB(II))
      CC(I)=.5*(AA(II)-BB(II))
```

```
      IF (ABS(CC(I))-SIG) 150,145,145
145   CONTINUE
      ISTOP=50
      GO TO 155
150   ISTOP=I
155   P=PHI
      P2=1.
      NQ=1
      IOS=1
      M2P=0
      I4=0
      ORELER=1.E25
      OR=1.E25
      DO 215 I=1,ISTOP
      PSAV(I)=P
      P2=P2*2.
      BD=TAN(P)*BB(I)/AA(I)
      BF=ATAN(BD)
160   INS=SIGN(1.,BF)
      IF (IOS*INS) 165,170,170
165   NQ=NQ+1
      IF (NQ.EQ.5) NQ=1
170   GO TO (175,190,190,195), NQ
175   IF (I4) 180,185,180
180   I4=0
      M2P=M2P+1
185   BE=BF+FLOAT(M2P)*TPI
      GO TO 200
190   BE=BF+PI+FLOAT(M2P)*TPI
      GO TO 200
195   BE=BF+TPI+FLOAT(M2P)*TPI
      I4=1
200   IOS=INS
      PR=P/BE
      RELER=ABS(OR-PR)/(PR+OR)
      IF (ORELER-RELER) 205,210,210
205   IOS=-IOS
      GO TO 160
210   P=BE+P
      OR=PR
215   ORELER=RELER
      TF=W*(PLUS+SIGNEM*(P/(P2*AA(ISTOP))))
      CALL TEK (0,RM,EK,EE)
      SUMEM=0.
      DO 220 IK=2,ISTOP
220   SUMEM=SUMEM+CC(IK)*SIN(PSAV(IK))
      TF=W*(PLUS1+SIGNEM*(EE/EK*TF+SUMEM))
      RETURN
      END
```

Appendix B:

Listing of subroutine TEK

```
      SUBROUTINE TEK (ID,RM,EK,E)
      DIMENSION RKP(60)
      IF (ID) 60,5,60
5     IF (RM-1.) 30,20,10
10    PRINT 15, RM
15    FORMAT (5X,9H*********,3X,13H LOOK OUT M =,F8.3,3X,9H*********)
      RETURN
20    EK=1.E75
      E=1.
25    RETURN
30    EK=1.57079632679489
      E=EK
      IF (RM) 10,25,35
35    IF (RM-.999) 40,40,65
40    RKN=SQRT(RM)
      DO 45 I=1,60
      RKP(I)=SQRT(1.-RKN*RKN)
      RKN=(1.-RKP(I))/(1.+RKP(I))
      IF (I.GE.2.AND.RKN.LT.1.E-20) GO TO 50
45    CONTINUE
      I=60
50    N=I-1
      DO 55 J=1,N
      T1=1.+RKP(I-J)
      EK=2.*EK/T1
55    E=T1*E-EK*RKP(I-J)
      RETURN
60    RPK=SQRT(RM)
      GO TO 70
65    RPK=SQRT(1.-RM)
70    PK2=RPK*RPK
      PKP=PK2
      GOL=ALOG(4./RPK)
      GK=GOL-1.
      FK=.25
      FE=.25
      EK=GOL+FK*GK*PKP
      E=1.+.5*(GOL-.5)*PKP
      GE=GK
      DO 85 I=2,2000
      R=FLOAT(I+I)
      D=R-1.
      PKP=PKP*PK2
      C=D/R
      FK=FK*D*D/(R*R)
      FE=FE*C
      H=1./(D*R)
      GK=GK-1./(D*FLOAT(I))
      GE=GE-H
      T1=FK*GK*PKP
      EK=T1+EK
      T2=FE*GE*PKP
      E=T2+E
      IF (T1-1.E-15) 75,75,80
75    IF (T2-1.E-15) 90,90,80
```

```
80    FE=FE*C                                                          TK  56
85    GE=GE-H                                                          TK  57
90    RETURN                                                           TK  58
      END                                                              TK  59
```